
BOITAOUTILS        (BOÎTE À OUTILS POUR J)         R.Coquidé (09/05/2018)

CONSTANTES

RC =: 10{a.
Fin de ligne avec retour chariot (passage à la ligne)

MD =: 10 58 10{a.
Séparation monad--dyad dans une programmation explicite

OUTILS GENERAUX

E =: 1 !:2&2
Affiche et retourne son unique argument de droite 
Ex : 55 + E (E 22* 56 67 78) – E 20 <. 12 56 7 

1232 1474 1716    
12 20 7
1220 1454 1709
1275 1509 1764

NAT =: 4!:0
Nature des objets. 
Retourne : _2 (invalide), _1 (inutilisé), 0 (nom), 1 (adverbe), 2 (conjonction), 3 (verbe)
Ex : NAT  <'Chrono'

3
NAT '*xx';'bonjour';'RC';'TABLE';'TA';'Feuler'

_2 _1 0 1 2 3

TYP =: 3!:0
Type de données
Retourne : 1 (booléen), 2 (caractère), 4 (entier), 8 (flottant), 16 (complexe), 32 (en boîte), 
64 (entier étendu), 128 (rationnel), 1024 (sparse booléen), 2048 (sparse caractère),       
4096 (sparse entier), 8192 (sparse flottant), 16384 (sparse complexe), 32768 (sparse en 
boîte), 65536 (symbol), 131072 (caractère2), 262144 (caractère4). 
Ex :    a =. 0 0 1 0 [ b =.'ZERT'[ c =. 45 
        d =. 3.14159 [ e =.4j8 [ f =. 234567x [ g =. 44r35

TYP f
64

TYP <f
32

TYP a
1

TYP b
2

TYP a;b;c
32

TYP &.> a;b;c;d;e;f;g
┌─┬─┬─┬─┬──┬──┬───┐
│1│2│4│8│16│64│128│
└─┴─┴─┴─┴──┴──┴───┘



P =: 9!:11
Fixe le nombre de chiffres affichés
Ex :       P  15
       1p1
   3.14159265358979

EFFACE =: 4!:55
Efface des objets en les détruisant (permet de libérer de la mémoire RAM)
 Ex : EFFACE <'LULU'  

1
 EFFACE 'RIRI';'JOJO';'MIMI'

1 1 1

Chrono =: 6!:2
Mesure du temps d’exécution d’une instruction (variable selon état de l’ordinateur)
Ex : Chrono 'W=.*:&.>1x+i. 1000000x'    NB. en secondes

1.22680 
Chrono 'W=.*:&.>1x+i. 1000000x'

2.25732
10 Chrono 'W=.*:&.>1x+i. 1000000x'

1.69680    NB. Moyenne sur 10 exécutions de l’instruction

CHAINES DE CARACTERES ET SUITES DE NOMBRES

SB =: (' '&~:#]) : (([~:])#])
Suppression des blancs dans une chaîne de car (utilisation monadique)
Ex1 : SB 'Bonjour chez vous'

Bonjourchezvous
Suppression d’un caractère précis dans une chaîne (utilisation dyadique)
Ex2 : 'Z' SB 'aZERtydZzZcvZfZZgZqsyb'

aERtydzcvfgqsyb
Suppression d’un nombre dans un vecteur numérique (utilisation dyadique)
Ex3 : 5 SB 6 5 7 5 5 4 2 5 9 5 8

6 7 4 2 9 8

Sauf =: ([:-.[ e.[:,])#[
Eléments du vecteur x sauf les éléments du vecteur y

SAUF =: [:/:~[:~.([:-.[e.[:,])#[
Idem avec suppression des dupliques et rangement dans l’ordre croissant

Ex : X=.12 45 34 6 3 5 7 34 12 5 7 9 62 [ Y=. 45 12 3 
X Sauf Y

34 6 5 7 34 5 7 9 62
[A=.2 3 $ 45 12 3 12 34 9

45 12 3
12 34 9

X Sauf A
6 5 7 5 7 62



X SAUF Y
5 6 7 9 34 62

X SAUF A
5 6 7 62

Minr =: (((}:<}.),1:)*.1:,}.<}:)#]
Minis relatifs d’une suite de nombres : R=. Minr Suite
Ex :  V=.55 3 8 9 12 15 8 6 9 2 4 6 0

Minr V
3 6 2 0

Maxr =: (((}:>}.),1:)*.1:,}.>}:)#]
Maxis relatifs d une suite de nombres  
Ex : V=.55 3 8 9 12 15 8 6 9 2 4 6 0

Maxr V
55 15 9 6

Pairs   =: (0:=2:|])#]
Extraction des nombres entiers pairs d une liste d’entiers

Impairs =: (1:=2:|])#]
Extraction des nombres entiers impairs d une liste d’entiers

Ex :  V=. 7 _8 5 6 32 9 1 _3 2 12 21 43 0 57
Pairs V

_8 6 32 2 12 0
Impairs V

7 5 9 1 _3 21 43 57

Delta =: 2&(-~/\)
Suite des différences des éléments d'une suite numérique
Ex :        [S=. (i.13)^4
0 1 16 81 256 625 1296 2401 4096 6561 10000 14641 20736 
             Delta S
1 15 65 175 369 671 1105 1695 2465 3439 4641 6095
 Delta Delta S        
14 50 110 194 302 434 590 770 974 1202 1454

Delta^:3 S
36 60 84 108 132 156 180 204 228 252

Delta^:4 S
24 24 24 24 24 24 24 24 24

Chg =: 4 : '($y)$(1{x)(((0{x)=,y)#i.$,y)},y'
Change l'élément 0{x en 1{x dans y 

Ex1 :   Ch=. 'abcdtfgsbcadbcbeacdg'
'c!' Chg  Ch 

ab!dtfgsb!adb!bea!dg



Ex2 : V=. 5 7 0 2 0 1 9 4 0 3
    0 _1 Chg V

5 7 _1 2 _1 1 9 4 _1 3

Chg travaille aussi avec des matrices

Ex3 : M;R=.0 _ Chg M=.4 5$?20$7 
┌─────────┬─────────┐
│5 5 0 4 5│5 5 _ 4 5│
│0 0 4 3 5│_ _ 4 3 5│
│1 6 5 2 1│1 6 5 2 1│
│1 0 1 5 3│1 _ 1 5 3│
└─────────┴─────────┘

Ex4 : M;R=.'*.' Chg M=.7 6$'az*ml*p*i**hg*'
┌──────┬──────┐
│az*ml*│az.ml.│
│p*i**h│p.i..h│
│g*az*m│g.az.m│
│l*p*i*│l.p.i.│
│*hg*az│.hg.az│
│*ml*p*│.ml.p.│
│i**hg*│i..hg.│
└──────┴──────┘

MANIPULATION DE MATRICES

SymH =: |.
Symétrique d une matrice (axe horizontal)

Ex : M;M1 =. SymH M=. ?3 5$9
┌─────────┬─────────┐
│2 2 6 1 7│3 5 2 7 3│
│0 2 8 7 5│0 2 8 7 5│
│3 5 2 7 3│2 2 6 1 7│
└─────────┴─────────┘

SymV =: |.&.|:
Symétrique d une matrice (axe vertical)

Ex : M;M1 =. SymV M
┌─────────┬─────────┐
│2 2 6 1 7│7 1 6 2 2│
│0 2 8 7 5│5 7 8 2 0│
│3 5 2 7 3│3 7 2 5 3│
└─────────┴─────────┘



SymC =: SymH @ SymV
Symétrique d une matrice (centre)

Ex : M;M1 =. SymC M
┌─────────┬─────────┐
│2 2 6 1 7│3 7 2 5 3│
│0 2 8 7 5│5 7 8 2 0│
│3 5 2 7 3│7 1 6 2 2│
└─────────┴─────────┘

Tr1 =: |:
Transposition (1ere diagonale) ou transposition classique

Ex : M;M1=.Tr1 M
┌─────────┬─────┐
│2 2 6 1 7│2 0 3│
│0 2 8 7 5│2 2 5│
│3 5 2 7 3│6 8 2│
│         │1 7 7│
│         │7 5 3│
└─────────┴─────┘

Tr2 =: |:&.|.
Transposition (2e diagonale)

Ex : M;M1=.Tr2 M
┌─────────┬─────┐
│2 2 6 1 7│3 5 7│
│0 2 8 7 5│7 7 1│
│3 5 2 7 3│2 8 6│
│         │5 2 2│
│         │3 0 2│
└─────────┴─────┘

Remarque : Les 5 précédents pro-verbes sont leur propre inverse.

Diag =: (<0 1)&|: :(([:(>:*i.)[:#])})
Lecture des éléments diagonaux d’une matrice 

Ex :             [M=.10 + ? 5 5$90
25 96 82 73 91
61 71 76 72 21
59 34 57 90 74
86 72 67 16 21
17 47 71 33 17

Diag M
25 71 57 16 17



Ecriture dans la diagonale d’une matrice

Ex1 :  W=.0 6 2 7 3
[M1=.W Diag M

0 96 82 73 91
61 6 76 72 21
59 34 2 90 74
86 72 67 7 21
17 47 71 33 3

Ex2 :
[M2=.0 Diag M

 0 96 82 73 91
61  0 76 72 21
59 34  0 90 74
86 72 67  0 21
17 47 71 33  0

Rotlig =: >@([|.&.><"1@])
Rotation sélective dans les lignes d une matrice 
 
Ex1 : M;M1 =.1 _2 3 Rotlig M=.i.3 5

┌──────────────┬──────────────┐
│ 0  1  2  3  4│ 1  2  3  4  0│
│ 5  6  7  8  9│ 8  9  5  6  7│
│10 11 12 13 14│13 14 10 11 12│
└──────────────┴──────────────┘

Ex2 : M;2 Rotlig M
┌──────────────┬──────────────┐
│ 0  1  2  3  4│ 2  3  4  0  1│
│ 5  6  7  8  9│ 7  8  9  5  6│
│10 11 12 13 14│12 13 14 10 11│
└──────────────┴──────────────┘

Rotcol =: Rotlig&.|:

Rotation sélective dans les colonnes d une matrice

Ex1 : M ;1 _1 0 2 0 Rotcol M
┌──────────────┬──────────────┐
│ 0  1  2  3  4│ 5 11  2 13  4│
│ 5  6  7  8  9│10  1  7  3  9│
│10 11 12 13 14│ 0  6 12  8 14│
└──────────────┴──────────────┘

Ex2 : M ;_2 Rotcol M 
┌──────────────┬──────────────┐
│ 0  1  2  3  4│ 5  6  7  8  9│
│ 5  6  7  8  9│10 11 12 13 14│
│10 11 12 13 14│ 0  1  2  3  4│
└──────────────┴──────────────┘



MATRICES DE BLOCS

Colenboite =: 3 : ('('''';1)<;.1 y',MD,'('''';x)<;.1 y')
Range les colonnes d une matrice dans des boites 
Ex1 : Colenboite M=. 10+?3 7$90 NB. Utilisation monadique

┌──┬──┬──┬──┬──┬──┬──┐
│27│88│41│46│25│82│50│
│50│40│46│69│66│47│76│
│20│40│71│91│98│73│90│
└──┴──┴──┴──┴──┴──┴──┘

Ex2 : 1 0 1 1 0 0 1 Colenboite M NB. Utilisation dyadique
┌─────┬──┬────────┬──┐
│27 88│41│46 25 82│50│
│50 40│46│69 66 47│76│
│20 40│71│91 98 73│90│
└─────┴──┴────────┴──┘

Ligenboite =: 3 : (',.(1;'''')<;.1 y',MD,',.x<;.1 y')
Range les lignes d une matrice dans des boîtes
Ex1 : Ligenboite M=.10+?5 7$90 NB. Utilisation monadique 

┌────────────────────┐
│79 42 44 95 91 66 77│
├────────────────────┤
│45 21 10 77 83 81 64│
├────────────────────┤
│44 97 29 68 76 26 22│
├────────────────────┤
│25 42 48 90 94 84 44│
├────────────────────┤
│18 48 36 14 44 79 98│
└────────────────────┘

Ex2 : 1 1 1 0 0 Ligenboite M NB. Utilisation dyadique
┌────────────────────┐
│79 42 44 95 91 66 77│
├────────────────────┤
│45 21 10 77 83 81 64│
├────────────────────┤
│44 97 29 68 76 26 22│
│25 42 48 90 94 84 44│
│18 48 36 14 44 79 98│
└────────────────────┘

MatBloc =: <;.1
Découpe une matrice en blocs
ex1 : MatBloc 10+?3 6$80 NB. Utilisation monadique
┌─────────────────┐
│21 47 25 56 83 24│
│87 39 11 49 57 10│
│59 37 14 63 46 33│
└─────────────────┘

NB. Utilisation dyadique



Ex2 :   [MB=.(1 0 1 0 0;1 0 0 1 1 0 1 0) MatBloc 10+?5 8$90
┌────────┬──┬─────┬─────┐
│68 53 64│30│51 44│79 10│
│31 50 56│49│12 97│58 50│
├────────┼──┼─────┼─────┤
│11 67 33│71│53 75│75 12│
│52 54 15│51│48 77│47 28│
│43 32 56│14│24 80│72 59│
└────────┴──┴─────┴─────┘

MatDbloc =: [: > 3 : ',"(#$y)&.>/y'^:(#@$)
Suprime les blocs d une matrice de blocs
Ex : MatDbloc MB
68 53 64 30 51 44 79 10
31 50 56 49 12 97 58 50
11 67 33 71 53 75 75 12
52 54 15 51 48 77 47 28
43 32 56 14 24 80 72 59
TABLEAUX DIVERS

TA =: 2 : ('[:',MD,'(n;,.x),.({.;}.)":y,x u/y')
Résultat d’un verbe mis sous forme d’un tableau

Ex1 :            HYPOTHENUSE =: +/&.*: 
3 5 7 HYPOTHENUSE TA 'HYP' 2 4 6 8

┌───┬───────────────────────────────┐
│HYP│2       4       6       8      │
├───┼───────────────────────────────┤
│ 3 │3.60555 5       6.7082  8.544  │
│ 5 │5.38516 6.40312 7.81025 9.43398│
│ 7 │7.28011 8.06226 9.21954 10.6301│
└───┴───────────────────────────────┘

Ex2 : V +. TA 'PGCD' V=.2 3 4 6 12 30 36 44 88
┌────┬──────────────────────┐
│PGCD│2 3 4 6 12 30 36 44 88│
├────┼──────────────────────┤
│  2 │2 1 2 2  2  2  2  2  2│
│  3 │1 3 1 3  3  3  3  1  1│
│  4 │2 1 4 2  4  2  4  4  4│
│  6 │2 3 2 6  6  6  6  2  2│
│ 12 │2 3 4 6 12  6 12  4  4│
│ 30 │2 3 2 6  6 30  6  2  2│
│ 36 │2 3 4 6 12  6 36  4  4│
│ 44 │2 1 4 2  4  2  4 44 44│
│ 88 │2 1 4 2  4  2  4 44 88│
└────┴──────────────────────┘



TAB =: 2 : ('[:',MD,'(x;(;:n)),((;:m),.<"0 y)')

Ex : M =. 3 4 $9 6 5 7 2 0 5 8 9 3 1 8
'Ventes' 'Eu Paris Aix' TAB 'Vis Clous Ecrous 

Boulons' M
┌──────┬───┬─────┬──────┬───────┐
│Ventes│Vis│Clous│Ecrous│Boulons│
├──────┼───┼─────┼──────┼───────┤
│Eu    │9  │6    │5     │7      │
├──────┼───┼─────┼──────┼───────┤
│Paris │2  │0    │5     │8      │
├──────┼───┼─────┼──────┼───────┤
│Aix   │9  │3    │1     │8      │
└──────┴───┴─────┴──────┴───────┘

TABLV =: ([:;:[),.[:<"0]
Ex : 'Vis Clous Boulons' TABLV 12.4 7.35 5.2
┌───────┬────┐
│Vis    │12.4│
├───────┼────┤
│Clous  │7.35│
├───────┼────┤
│Boulons│5.2 │
└───────┴────┘
TABLH =: ([:;:[),:[:<"0 ]

Ex : 'A B C D' TABLH 4 15 308 7
┌─┬──┬───┬─┐
│A│B │C  │D│
├─┼──┼───┼─┤
│4│15│308│7│
└─┴──┴───┴─┘

TABLE =: 1 : (':'; '(((#~LF-.@e.])5!:5<''u'');,.y),. 
({.; }.)":x,y u/x')~

Ex1 : HYP =: +&.*:   NB. hypthénuse  ( avec P 4 )
3 5 9 HYP TABLE 2 3 4 5 6

┌───┬─────────────────────────────┐
│HYP│    2     3     4     5     6│
├───┼─────────────────────────────┤
│3  │3.606 4.243 5     5.831 6.708│
│5  │5.385 5.831 6.403 7.071 7.810│
│9  │9.220 9.487 9.849 10.30 10.82│
└───┴─────────────────────────────┘
Ex2 : 0 1 *. TABLE 0 1 NB. Table de vérité du « et »
┌──┬───┐
│*.│0 1│
├──┼───┤
│0 │0 0│
│1 │0 1│
└──┴───┘

mailto:#~LF-.@e.])5!:5%3C''u'');,.y)


TABBLOC =: 2 : ('[:',MD,'((<'' ''),(;:n)),(;:m),.x <;.1 y')

Ex : (1 0 1 1 0;1 1 0 0 1 1 0) ('Eu Paris Aix' TABBLOC 
'vis boulons ecrous clous') ?5 7$99
┌─────┬───┬────────┬──────┬─────┐
│     │vis│boulons │ecrous│clous│
├─────┼───┼────────┼──────┼─────┤
│Eu   │46 │ 9 37 23│68    │89 78│
│     │ 2 │81 52 59│ 6    │ 0 93│
├─────┼───┼────────┼──────┼─────┤
│Paris│ 1 │25 88 73│19    │ 0 84│
├─────┼───┼────────┼──────┼─────┤
│Aix  │13 │94 11 37│20    │61 75│
│     │ 9 │75 14 53│92    │17 73│
└─────┴───┴────────┴──────┴─────┘

Factor=:3 : '~./:~((#:i.@(*/))#~#q)<@(/:~)@(*//.)"1 q=.q: y'
Décomposition d’un nombre en produits de facteurs
Ex : Factor 105
┌─────┬────┬────┬────┬───┐
│3 5 7│3 35│5 21│7 15│105│
└─────┴────┴────┴────┴───┘

Pro-conjonctions (divers produits de composition de 2 verbes ou pro-verbes)
    Utilisation monadique
UVY   =: 2 : 'u@v' NB. u(v y)
VUY   =: 2 : 'v@u'      NB. v(u y)
UY_VY =: 2 : '([:u])v]' NB. (u y)v y
VY_UY =: 2 : '([:v])u]' NB. (v y)u y
U_YVY =: 2 : 'u@(]v])'   NB. u(y v y)
V_YUY =: 2 : 'v@(]u])' NB. v(y u y)
YU_VY =: 2 : ']u[:v]'    NB. y u (v y)
YV_UY =: 2 : ']v[:u]'  NB. y v (u y)
    Utilisation dyadique
XUVY  =: 2 : 'u v' NB. x u (v y)
YUVX  =: 2 : '(u v)~' NB. y u (v x)
UY_VX =: 2 : 'v~u'       NB. (u y) v x
UX_VY =: 2 : '(v~u)~' NB. (u x) v y
VY_UX =: 2 : 'u~v' NB. (v y) u x
VX_UY =: 2 : '(u~v)~' NB. (v x) u y
UXVY  =: 2 : 'u@v' NB. u (x v y)
VXUY  =: 2 : 'v@u' NB. v (x u y)
UYVX  =: 2 : 'u@(v~)' NB. u (y v x)
VYUX  =: 2 : 'v@(u~)' NB. v (y u x)

Ex ; X =. 6 [ Y =. 5
(% UVY ^) Y

0.00673795



(% YV_UY ^) Y
1.37973

X (% UY_VX ^) Y
6.4e_5

X (% VX_UY ^) Y
80.6858

ARRONDIS

Flot =: x:^:_1
Ecriture approximative en flottant de nombres entiers étendus ou rationnels  (ici avec  P  8)
Ex : Flot _73r119 1234567891213141517856x

_0.61344538 1.2345679e21

Approximations :
arpe=:(0 arpe]) :((10^[)*[:>.]%10^[)"0 0 0      NB. Par excès
arpd=:(0 arpd]) :((10^[)*[:<.]%10^[)"0 0 0   NB. Par défaut
arpp=:(0 arpp]) :((10^[)*[:<. 0.5+]%10^[)"0 0 0 NB. Au plus près
Ex1 : 2 chiffres après le point décimal 

_2 arpp 53.96543 _6.90743  8.7245       NB. Au plus près
53.97 _6.91 8.72

_2 arpe 53.96543 _6.90743  8.7245   NB. Par excès
53.97 _6.9 8.73

_2 arpd 53.96543 _6.90743  8.7245   NB. Par défaut
53.96 _6.91 8.72

Ex2 : 2 chiffres avant le point décimal (multiples de 100)
2 arpp 4567.8 _65432.965 12345.89        NB. Au plus près

4600 _65400 12300
2 arpe 4567.8 _65432.965 12345.89        NB. Par excès

4600 _65400 12400
2 arpd 4567.8 _65432.965 12345.89   NB. Par défaut 

4500 _65500 12300
Ex3 : par valeurs entières

0 arpp 4567.8 _65432.965 12345.89        NB. Au plus près
4568 _65433 12346

0 arpe 4567.8 _65432.965 12345.89        NB. Par excès
4568 _65432 12346

0 arpd 4567.8 _65432.965 12345.89   NB. Par défaut
4567 _65433 12345

Comme = avec tolérance (m entier)

Absolue 10^m
EGTA =: 1 : ('[:',MD,'0=<.(10^-m)*|x-y')

 Relative 10^m
EGTR =: 1 : ('[:',MD,'0=<.(10^-m)*(|x-y)%(|x)<.(|y)')



Ex: 50 (3 EGTA) 5 0.000005 0.05 50 500 5000 5000000 
1 1 1 1 1 0 0 

Ex : 50 (3 EGTR) 5 0.000005 0.05 50 500 5000 5000000 
1 0 1 1 1 1 0 

coupe =: sint =: {.@]+-/@|.@]*i.@>:@[%[
Découpage de l intervalle y en x sous-intervalles égaux
Ex : 10 sint 0 2

0 0.2 0.4 0.6 0.8 1 1.2 1.4 1.6 1.8 2
SURF =: 11&o.@-:@(+ +/ .* 1&|.) :([:SURF[j.])
Surface algébrique délimitée par un polygone plan défini par ses sommets
Ex1 : 4 6 8 9 7 5 2 3 SURF 2 3 5 7 6 4 3 1

11
SURF 4j2 6j3 8j5 9j7 7j6 5j4 2j3 3j1

11
Ex2 :  4.56 7.89 9.54 4.12 2.34 SURF 1.2 7 5.3 8.04 5.76

16.9073
SURF 4.56j1.2 7.89j7 9.54j5.3 4.12j8.04 2.34j5.76

16.9073

QUELQUES FONCTIONS CLASSIQUES

Feuler =: !@<: :(*&$:%$:@+)

Fonctions gamma (forme monadique) et beta (forme dyadique) d'EULER
1

0

1 1 1

0

( )   ; 0 (s'étend à t -{0 _1 _2 _3 ...})

( , ) (1 )   ; 0 et 0 (s'étend à  et )

z t

p q

Gamma t e z dz t E

Beta p q z z dz p q p E q E

∞ − −

− −

= > ∈ =

= − > > ∈ ∈

∫
∫

£

Ex1 : Calcul de Gamma(5),Gamma(5.7531),Gamma(_3.2),Gamma(1j2)
Feuler  5  5.7531  _3.2  1j2

24 79.191 0.689056 0.151904j0.0198049

Ex2 : Calcul de Beta(3,5),Beta(5.2,_3.7),Beta(1j3,5j2) 
3  5.2  1j3  Feuler  5  _3.7  5j2

0.00952381 9.25055 0.0096253j0.037409

Fgauss =: 1r2p1&%@{.@(((+/%#)&.:^.,+/%#)^:_)

La fonction de GAUSS 
( , )  où 0 , 0 

2 ( , )

( , )=Moyenne arithmético-géométrique de a et b   

ag

ag

Fgauss a b a b
M a b

M a b

π= > >
  

est utile pour calculer certaines intégrales :



2 2 2 2 2 2 2 2 2 2
0 0 0

4 2 2 2 4 2 2 2
0 0

1
2 ( , ) où 0,  0

2( )( ) ( )( ) ( )( )

1
2 ( 2 , ) où | | 2

2) )

ab

p

dt dt dt
Fgauss a b a b

t a t b t a t b t t a t b

dt dt
Fgauss p s p s p

t s t p t s t p

∞ ∞

∞

= = = > >
+ + + + + +

= = + <
+ + + +

∫ ∫ ∫

∫ ∫
Ex : Calcul de   Fgauss(a,b) 

Fgauss 3 5.8
0.366588570138

Fgauss 0.35 9.82
0.480827


